Jednovýběrové testy

Komentované řešení pomocí MS Excel

Vstupní data

	A	B	C
1		Naměřené koncentrace olova	
2	Pořadí měření	Potrubí 1.	Potrubí 2.
3	1	10,18	9,6
4	2	9,42	10,2
5	3	9,05	10,42
6	4	9,07	8,62
7	5	9,95	9,47
8	6	9,35	10,41
9	7	10,21	9,96
10	8	8,93	10,04
11	9	9,21	8,91
12	10	8,76	9,86
13	11	9,6	9,16
14	12	10,23	9,6
15	13	9,61	10,66
16	14	9,13	11,22
17	15	9,48	11,12
18	16	10,61	9,93
19	17	9,18	10,71
20	18	9,54	8,39
21	19	9,57	8,58
22	20	9,04	11,3
23	21	10	9,36

- V dalším budeme předpokládat, že tabulka se vstupními daty je umístěna v oblasti A1:C23 (viz. obrázek)

Základní statistiky

	A	B	C
21	**19**	9,57	8,58
22	**20**	9,04	11,3
23	**21**	10	9,36
24			
25	Prumer	=PRŮMĚR(B3:B23)	
26	S	=SMODCH.VÝBĚR(B3:B23)	
27	n	=POČET(B3:B23)	

- vložíme vzorce pro výpočet výběrového průměru, výběrové směrodatné odchylky (S) a počtu měření (n)

	A	B	C
1		**Naměřené koncentrace**	
2	**Pořadí měření**	**Potrubí 1.**	**Potrubí 2.**
24			
25	Prumer	9,53	9,88
26	S	0,50	0,86
27	n	21	21

- Vzorce zkopírujeme i do sloupce „C“. Vidíme, že průměrná koncentrace v obou potrubích je menší než 10 μg/l. Budeme testovat, zda je tento rozdíl statisticky významný. Variabilita koncentrací v prvním potrubí je menší než variabilita ve druhém potrubí.

Ověření předpokladů – střední hodnoty

Postup:

- K tabulce přidáme sloupec s limitními hodnotami 10 (D3:D23) – pro zobrazení čáry odpovídající limitu.
- Označíme oblast s měřeními včetně hlavičky a přidaného limitu (B2:D23) → otevřeme kartu „Vložení“ → klikneme na „Spojnicový (panel „Grafy“) a vybereme „spojnicový se značkami“.
- Pro větší přehlednost:
  - Upravíme formát datové řady představující limit 10 a odstraníme jí z legendy,
  - Přidáme názvy os a název grafu.



	A	B	C	D
1		Naměřené koncentrace olova		
2	Pořadí měření	Potrubí 1.	Potrubí 2.	Limit
3	1	10.18	9.6	10
4	2	9.42	10.2	10
5	3	9.05	10.42	10
6	4	9.07	8.62	10
7	5	9.95	9.47	10
8	6	9.35	10.41	10
9	7	10.21	9.96	10
10	8	8.93	10.04	10
11	9	9.21	8.91	10
12	10	8.76	9.86	10
13	11	9.6	9.16	10
14	12	10.23	9.6	10
15	13	9.61	10.66	10
16	14	9.13	11.22	10
17	15	9.48	11.12	10
18	16	10.61	9.93	10
19	17	9.18	10.71	10
20	18	9.54	8.39	10
21	19	9.57	8.58	10
22	20	9.04	11.3	10
23	21	10	9.36	10



Interpretace výsledků

- V grafu vývoje naměřených koncentrací není patrný žádný trend. **Lze** tedy **předpokládat**, že **střední hodnoty** jednotlivých měření **jsou** v případě obou potrubí **konstantní** (tj. nezávisí na čase).
- Dále můžeme pozorovat, že naměřená koncentrace olova u obou potrubí v některých případech překročila limitní mez 10 µg/l.
- U potrubí 2 došlo k překročení častěji a o větší hodnoty než u potrubí 1.

Ověření předpokladů – rozptyly

<u>Postup:</u>

- Pro každé potrubí spočítáme rezidua ($e = X_i - \bar{X}$) v nové tabulce: vložíme vzorec „=B3-B$25“ do buňky I29 a rozkopírujeme do celé oblasti I29:J49
- Označíme oblast s rezidui včetně hlavičky (I28:J49) → otevřeme kartu „Vložení“ → klikneme na „Sloupcový (panel „Grafy“) a vybereme „Skupinový sloupcový“
- Pro větší přehlednost:
  - Na kartě „Návrh“ klikneme na „Zaměnit řádek za sloupec“ a na téže kartě změníme styl grafu na jednobarevný,
  - Odebereme legendu, přidáme název svislé osy a název grafu.

	H	I	J
27	**Rezidua**		
28	Pořadí měření	Potrubí 1.	Potrubí 2.
29	1	=B3-B$25	-0.281904762
30	2	-0.10952381	0.318095238
31	3	-0.47952381	0.538095238
32	4	-0.45952381	-1.261904762
33	5	0.42047619	-0.411904762
34	6	-0.17952381	0.528095238
35	7	0.68047619	0.078095238
36	8	-0.59952381	0.158095238
37	9	-0.31952381	-0.971904762
38	10	-0.76952381	-0.021904762
39	11	0.07047619	-0.721904762
40	12	0.70047619	-0.281904762
41	13	0.08047619	0.778095238
42	14	-0.39952381	1.338095238
43	15	-0.04952381	1.238095238
44	16	1.08047619	0.048095238
45	17	-0.34952381	0.828095238
46	18	0.01047619	-1.491904762
47	19	0.04047619	-1.301904762
48	20	-0.48952381	1.418095238
49	21	0.47047619	-0.521904762



<u>Interpretace výsledků</u>

- V grafech vývoje vypočítaných reziduí jednotlivých potrubí není patrný žádný systematický vývoj tedy nedochází ani k výraznému růstu ani poklesu absolutních hodnot reziduí. Můžeme tedy předpokládat, že **rozptyly** jednotlivých měření **jsou** v případě obou potrubí **konstantní** (tj. nezávisí na čase).
- Rezidua se jeví velmi náhodně a nesystematicky, což je důležité pro aplikaci jednovýběrových testů.

Ověření normality – histogramy 1

Postup:

- Měření musíme rozdělit do intervalů.
  - Počet intervalů (k) určíme podle Yuleova pravidla: $k = 2{,}5*n^{1/4}$
  - Šířka intervalu bude $d = (Max - Min)/k$, kde s ohledem na předpokládanou normalitu volíme $Max = Prumer + 2*S$, a $Min = Prumer - 2*S$

	H	I	J
52	počet intervalů (yuleovo pravidlo)	=ROUNDDOWN(2,5*POWER(B27;1/4);0)	5
53	min	=B$25-2*B$26	8,16
54	max	=B$25+2*B$26	11,61
55	sirka intervalu	=(I54-I53)/I52	0,69

  - Nyní sestavíme intervaly: (0,Min),(Min, Min+d),…,(Max –d, Max), (Max, 100) a spočítáme středy intervalů (střed prvního a posledního intervalu volíme jako příslušná krajní hodnota ± d/2).
- Spočítáme relativní četnosti měření v jednotlivých intervalech (počet měření v daném intervalu / celkový počet měření) – pomocí funkce **COUNTIF**
- Pro srovnání dopočítáme očekávané pravděpodobnosti v jednotlivých intervalech za předpokladu normality (pomocí distribuční funkce normálního rozdělení **NORMDIST**, kde zvolíme $\mu = Prumer$, a $\sigma = S$.
- Kromě funkce COUNTIF by se tabulka četností dala sestrojit také pomocí nástroje „Kontingenční tabulka“.

	H	I	J	K	L
57	**Intervaly koncentrace - Potrubí 1.**				
58	Dolní mez	Horní mez	Střed	Relativní četnost	Očekávaná pravděpodobnost
59	0,00	=$I$53	=I59-$I$55/2	0,000	0,02
60	=I59	=H60+$I$55	=(H60+I60)/2	0,048	0,092
61	8,93	9,33	9,13	0,333	0,230
62	9,33	9,73	9,53	0,333	0,311
63	9,73	10,13	9,93	0,095	0,230
64	10,13	10,53	10,33	0,143	0,092
65	10,53	100,00	10,73	0,048	0,023

	H	I	J	K	L
68	**Intervaly koncentrace - Potrubí 2.**				
69	Dolní mez	Horní mez	Střed	Relativní četnost	Očekávaná pravděpodobnost
70	0,00	8,16	7,81	0,000	0,023
71	8,16	8,85	8,50	0,143	0,092
72	8,85	9,54	9,19	0,190	0,230
73	9,54	10,23	9,88	0,333	0,311
74	10,23	10,92	10,57	0,190	0,230
75	10,92	11,61	11,26	0,143	0,092
76	11,61	100,00	11,95	0,000	0,023

	K
59	=COUNTIFS($B$3:$B$23;">="&H59;$B$3:$B$23;"<"&I59)/$B$27

	L
59	=NORMDIST(I59;$B$25;$B$26;1)-NORMDIST(H59;$B$25;$B$26;1)

Ověření normality – histogramy 2

Postup:

- Z tabulek relativních četností (viz. předchozí slide) dále vytvoříme sloupcové grafy. Postup pro oba grafy je podobný.
  - Označíme tabulku s relativními četnostmi a očekávanými pravděpodobnostmi (K58:L65)
  - Karta „Vložení“ → „Sloupcový“ (panel „Grafy“) → vybereme „Skupinový sloupcový“
  - Pro přehlednost přidáme název grafu, název vodorovné osy a do popisků vodorovné osy vložíme středy intervalů.





Interpretace výsledků:

- U Potrubí 2 jsou rozdíly mezi relativními četnostmi a očekávanými pravděpodobnostmi (za předpokladu normality) poměrně malé a předpoklad normality je tedy oprávněný. V případě Potrubí 1 jsou tyto rozdíly větší, ovšem vzhledem k relativně malému počtu pozorování to ještě nemusí nutně znamenat, že rozdělení koncentrací není normální.
- Potvrzují se naše předchozí zjištění o tom, že koncentrace v Potrubí 2 jsou ve srovnání s Potrubím 1 o trochu vyšší a více rozptýlené.
- Výsledná podoba histogramu může být velmi citlivá na volbu intervalů (jsou-li naměřené hodnoty blízko hranicím intervalů, může i při malé změně hranic dojít k zásadní změně v histogramu). Protože je předpoklad normality klíčový, ověříme jej ještě pomocí tzv. Q-Q grafů.

Ověření normality – Q-Q grafy 1

Postup:

- Naměřené hodnoty v jednotlivých potrubích musíme nejdříve vzestupně uspořádat:
  - Zkopírujeme naměřené hodnoty do nové oblasti (jinak bychom ztratili původní uspořádání).
  - Označíme koncentrace pro Potrubí 1 (vč. hlavičky), tj. I88:I109.
  - Karta „Data“ → „Seřadit“ (panel „Seřadit a filtrovat“) → vybereme „Pokračovat s aktuální oblastí“ → stiskneme „Seřadit“→ ponecháme nastavení Seřadit podle: Potrubí 1; Řazení: Hodnoty; Pořadí: Od nejmenšího k největšímu.
  - Podobně postupujeme pro Potrubí 2.
- Dále dopočítáme
  - kumulativní pravděpodobnosti s korekcí na spojitost (pro *i*-tou nejmenší hodnotu je to *(i-0,5)/n*,
  - příslušné teoretické kvantily normovaného normálního rozdělení N(0,1) , tedy $u_i = \Phi^{-1}\,[(i-0,5)/n]$, a to pomocí funkce **NORMSINV,**
  - Pro každé potrubí zvlášť: standardizované hodnoty uspořádaných koncentrací, tj. *Y=(X–Prumer)/S* , které představují skutečné (napozorované) kvantily.

	K
89	=(H89-0,5)/POČET($H$89:$H$109)

	L
89	=NORMSINV(K89)

	M
89	=(I89-PRŮMĚR(I$89:I$109))/SMODCH(I$89:I$109)

	N
89	=(J89-PRŮMĚR(J$89:J$109))/SMODCH(J$89:J$109)

	H	I	J	K	L	M	N
87	Pořadí	Uspořádané koncentrace		Kumulativní	Kvantil	Stadnardizované koncentrace	
88	hodnoty	Potrubí 1.	Potrubí 2.	pravděp.	N(0,1)	Potrubí 1.	Potrubí 2.
89	1	8,76	8,39	0,024	-1,981	-1,574	-1,771
90	2	8,93	8,58	0,071	-1,465	-1,226	-1,545
91	3	9,04	8,62	0,119	-1,180	-1,001	-1,498
92	4	9,05	8,91	0,167	-0,967	-0,981	-1,153
93	5	9,07	9,16	0,214	-0,792	-0,940	-0,857
94	6	9,13	9,36	0,262	-0,637	-0,817	-0,619
95	7	9,18	9,47	0,310	-0,497	-0,715	-0,489
96	8	9,21	9,6	0,357	-0,366	-0,654	-0,335
97	9	9,35	9,6	0,405	-0,241	-0,367	-0,335
98	10	9,42	9,86	0,452	-0,120	-0,224	-0,026
99	11	9,48	9,93	0,500	0,000	-0,101	0,057
100	12	9,54	9,96	0,548	0,120	0,021	0,093
101	13	9,57	10,04	0,595	0,241	0,083	0,188
102	14	9,6	10,2	0,643	0,366	0,144	0,378
103	15	9,61	10,41	0,690	0,497	0,165	0,627
104	16	9,95	10,42	0,738	0,637	0,860	0,639
105	17	10	10,66	0,786	0,792	0,962	0,923
106	18	10,18	10,71	0,833	0,967	1,331	0,983
107	19	10,21	11,12	0,881	1,180	1,392	1,469
108	20	10,23	11,22	0,929	1,465	1,433	1,588
109	21	10,61	11,3	0,976	1,981	2,210	1,683

Ověření normality – Q-Q grafy 2

Postup:

- Označíme oblast dat s teoretickými kvantily N(0,1) a standardizovanými uspořádanými koncentracemi (L89:M109).
- Karta „Vložení“ → „Bodový“ (panel „Grafy“) → vybereme „Bodový pouze se značkami“.
- Přidáme novou řadu (diagonálu): Karta „Návrh“→ „Vybrat data“ → „Přidat“ (Položky legendy(řady)) → Hodnoty X řad: vložíme oblast s kvantily N(0,1), tedy L89:L109; Hodnoty Y řad: vložíme totéž (L89:L109).
- Upravíme formát nově přidané řady (pravým tlačítkem klikneme na příslušnou řadu v grafu): Možnosti značek = vybereme „žádné“, Barva čáry = vybereme „plná čára“ a zvolíme červenou barvu.
- Pro přehlednost přidáme název grafu a odebereme legendu.





Interpretace výsledků:

- Q-Q graf vykresluje teoretické (očekávané) kvantily (osa X) vůči empirickým (naměřeným) kvantilům (osa Y). Čím blíže jsou jednotlivé body červené ose 1.a 3. kvadrantu, tím lépe empirické rozdělení odpovídá tomu teoretickému.
- Na rozdíl od histogramu zde nedochází ke zkreslení kvůli zařazení dat do intervalů, na druhou stranu Q-Q graf neznázorňuje polohu ani variabilitu pozorování.
- V obou případech (Potrubí 1 i Potrubí 2) jsou jednotlivé body velmi blízko červené ose 1. a 3. kvadrantu, což indikuje, že empirické rozdělení odpovídá rozdělení teoretickému. Můžeme tedy oprávněně předpokládat, že naměřené koncentrace u obou potrubí pochází z normálního rozdělení. To není velkým překvapením, protože chyby v měření se velmi často řídí normálním rozdělením.

Test o střední hodnotě (úloha A)

Postup (podrobný popis a vzorce viz. *teorie-jednovyberovy test.pdf*):

- Nastavíme střední hodnotu pro nulovu hypotézu ($\mu_0 = 10$).
- Spočítáme testovou statistiku.
- Podle alternativní hypotézy (H_1: $\mu<\mu_0$) a hladiny významnosti spočítáme kritickou hodnotu a to pomocí funkce **TINV**.
  - Pozor, TINV není inverzní funkce k distribuční funkci t-rozdělení, je to spíše inverzní funkce k oboustranné funkci přežití tohoto rozdělení, tedy je-li *x=TINV(p,k)*, potom $p = P(|X| > x)$, kde X~t(k). Více viz. nápověda k této funkci. Porovnáním testové statistiky s kritickou hodnotou vyhodnotíme test.
- Pro přesnější představu o průkaznosti výsledků spočítáme p-hodnotu pomocí funkce **TDIST**.
  - Pozor, ani TDIST není distribuční funkce t-rozdělení, nýbrž funkce přežití (buď jednostranná nebo oboustranná dle nastavení parametru „strany“), tedy $TDIST = P(X>x)$ nebo $TDIST = P(|X| > x)$. Více viz. nápověda k této funkci.

	A	B	C
35	**Úloha A**	**Potrubí 1.**	**Potrubí 2.**
36	nulova hypoteza	10	10
37	hl. významnosti	0,05	0,05
38	Testova statistika	-4,30403	-0,62681
39	Kriticka hodnota	-1,72472	-1,72472
40	**zamítáme H_0**	**ANO**	**NE**
41	**p-hodnota**	**0,00017**	**0,26894**

	B
36	10
37	0,05
38	=((B25-B36)/B26)*ODMOCNINA(B27)
39	=-TINV(2*B37;B27-1)
40	**=KDYŽ(B38<B39;"ANO";"NE")**
41	**=TDIST(-B38;B27-1;1)**

Interpretace výsledků (obecný popis viz. *teorie-jednovyberovy test.pdf*):

- **V případě Potrubí 1**. zamítáme H_0: $\mu=\mu_0$ a přijímáme alternativní hypotézu H_1: $\mu<\mu_0$. To znamená, že naměřené koncentrace prokázaly, že **střední koncentrace olova je nižší než stanovený limit 10µg/l**. Pravděpodobnost chybného závěru je dána hl. významnosti, tj. 5 %. Spočítaná p-hodnota (minimální hl. významnosti pro zamítnutí H_0) je hluboko pod 5 %, takže naměřené výsledky jsou opravdu velmi průkazné.
- **V případě Potrubí 2.** nulovou hypotézu H_0: $\mu=\mu_0$ nezamítáme. Naměřené výsledky jsou tedy neprůkazné a **střední koncentrace olova nemusí být nižší než stanovený limit 10µg/l**. To ovšem nemusí znamenat, že střední koncentrace je vyšší než limit (vždyť naměřený průměr je 9,88, což je stále pod limitem). To pouze znamená, že nemůžeme učinit žádný přesvědčivý závěr. Dosažená p-hodnota výrazně převyšuje 5%, tudíž výsledky jsou opravdu velmi neprůkazné. Ve srovnání s Potrubím 1.jsou totiž naměřené hodnoty koncentrace o něco vyšší (průměr je blíže limitu) a je zde i vyšší variabilita naměřených dat, což zvyšuje míru nejistoty ohledně skutečné střední hodnoty.

Test o rozptylu (úloha B)

Postup (podrobný popis a vzorce viz. *teorie-jednovyberovy test.pdf*):

- Pro lepší představu odhadneme variační koeficient (V) z dat.
- Spočítáme mezní směrodatnou odchylku odpovídající nulové hypotéze (σ_0 = 0,08*Prumer).
- Spočítáme testovou statistiku.
- Protože nemáme zadanou požadovanou hladinu významnosti, nebudeme počítat kritický obor.
- Spočítáme p-hodnotu pomocí funkce **CHIDIST**.
  - Pozor, CHIDIST není distribuční funkce χ^2-rozdělení, nýbrž funkce přežití tohoto rozdělení, tedy *CHIDIST = P(X>x)*. Více viz. nápověda k této funkci.

	A	B	C
45	**Úloha B**	**Potrubí 1.**	**Potrubí 2.**
46	odhad V	**5,26%**	**8,74%**
47	nulova hypoteza	0,76	0,79
48	Testova statistika	8,6	23,9
49	**p-hodnota**	**0,013**	**0,751**

	B
46	=B26/B25
47	=0,08*B25
48	=+(B27-1)*B26^2/B47^2
49	=1-CHIDIST(B48;B27-1)

Interpretace výsledků (obecný popis viz. *teorie-jednovyberovy test.pdf*):

- **V případě Potrubí 1**. jsme dosáhli velmi nízké p-hodnoty = 0,013. Můžeme tedy zamítnout H_0: $\sigma = \sigma_0$ ve prospěch H_1: $\sigma < \sigma_0$ a to i pro hl. významnosti 1,3% . Na základě naměřených koncentrací tedy můžeme **se spolehlivostí vyšší než 98%** (tj. s pravděpodobností chyby méně než 2 %) tvrdit, že předložený **výběr je získán s přesností lepší než 8 %** . Toto **tvrzení lze tedy považovat za prokázané**. Tomu odpovídá i odhadnutá hodnota variačního koeficientu 5,26%, což je výrazně pod požadovanými osmi procenty.
- **V případě Potrubí 2**. vychází velmi vysoká p-hodnota = 0,751. Pro zamítnutí H_0: $\sigma = \sigma_0$ ve prospěch H_1: $\sigma < \sigma_0$ by hl.významnosti musela být alespoň 76 %. Tvrdit, že předložený **výběr je získán s přesností lepší než 8 %** můžeme tedy **pouze se spolehlivostí mensí než 25%** (tj. s pravděpodobností chyby více než 75 %). Toto **tvrzení tedy** rozhodně **nelze považovat za prokázané**. Tomu odpovídá i odhadnutá hodnota variačního koeficientu 8,74%, což je dokonce více než požadovaných osm procent. Pozor, nemusí to ještě automaticky znamenat, že výběr je získán s přesností horší než 8 %. Toto můžeme tvrdit se spolehlivostí cca 75 % (tj. pravděpodobností chyby 25 %), což je běžně považováno za neprůkazné.